
Abstract
We present a cooperative thread library for Objective Caml.The
library is entirely written in Objective Caml and does not rely on
any external C function. Programs involving threads are written in
a monadic style. This makes it possible to write threaded code al-
most as regular ML code, even though it has a different semantics.
Cooperative threads are especially well suited for concurrent net-
work applications, where threads perform little computation and
spend most of their time waiting for input or output, at whichtime
other threads can run. This library has been successfully used in the
Unison file synchronizer and the Ocsigen Web server.

Categories and Subject Descriptors D.1.1 [PROGRAMMING
TECHNIQUES]: Applicative (Functional) Programming; D.1.3
[PROGRAMMING TECHNIQUES]: Concurrent Programming;
D.3.3 [PROGRAMMING LANGUAGES]: Language Constructs
and FeaturesConcurrent programming structures

General Terms Design, Languages

Keywords Thread, Concurrency, Networking, Programming, Im-
plementation, Monad, Objective Caml, ML

1. Introduction
The Lwt (“lightweight threads”) library is a cooperative thread
library for Objective Caml (Leroy et al.): threads have to explicitly
yield the control to other threads rather than being preempted by
a scheduler. The motivations for this library are two-fold.First,
we believe that cooperative threads are easier to program with
than preemptive threads. Indeed, as context switches are explicit
(typically through a call to ayield function or due to a blocking
system call), there are much less opportunities for race conditions.
Second, compared to system threads, the threads are extremely
lightweight: they only hold the values required for the remaining of
the computation rather than a full stack. Therefore, there can easily
be thousands of them. Context switch is also cheaper as it consists
in just a few function calls. Cooperative threads are especially well
suited for highly concurrent network applications (von Behren et al.
2003), where threads perform little computation and spend most of
their time waiting for input or output, at which time other threads
can run.

The library is entirely written in Objective Caml, contraryto
the standard thread libraries (system threads and VM-levelthreads)
which rely on external C functions. Programs involving threads are
written in a monadic style. This makes it possible to write threaded
code almost as regular ML code, even though it has a different
semantics.

This library has been successfully used since 2002 in the Unison
file synchronizer (Balasubramaniam and Pierce 1998; Pierceand
Vouillon 2004) for the simultaneous synchronization of several

files. It is now used in the Ocsigen Web server (Balat 2006). The
library is available online1. It is also included in the Debian Linux
distribution.

We start with simple examples giving a feel of the library (sec-
tion 2). We detail the library API (section 3). We provide more
complex examples demonstrating how to use the library (section 4).
Finally, we describe the implementation (section 5).

2. A quick presentation of the library
In order to give a feel of the library, we present two simple ex-
amples. With the first example, we explain how to write a func-
tion performing I/Os asynchronously. The second one shows two
threads running concurrently.

First, consider the following function. It reads up to one kilobyte
from file descriptorin_fd, sleeps for three seconds and writes the
data to file descriptorout_fd. Any of the three system calls may
block, interrupting for some time the whole program.

let forward in_fd out_fd buffer =
let n = Unix.read in_fd buffer 0 1024 in
Unix.sleep 3;
let n’ = Unix.write out_fd buffer 0 n in
()

The core idea of the library is to replace a possibly blockingfunc-
tion by a function that returns immediately apromise, that is a
value that acts as a proxy for the value eventually computed by the
function. Thus, instead of functionUnix.sleep, one uses function
Lwt_unix.sleep of type int -> unit Lwt.t whereLwt.t is
the type of promises. Likewise, the functionUnix.read is replaced
by a functionLwt_unix.read of type

file_descr -> string -> int -> int -> int Lwt.t

Type Lwt.t can also be interpreted as the type ofthreads: the
functionLwt_unix.read returns a thread that reads from a given
file descriptor and terminates with return value the number of byte
read. We adopt this interpretation for the remainder of the paper.

The library provides basic functions for handling threads.The
function Lwt.bind is used to perform an action when the return
value of a thread becomes available: the expression

Lwt.bind e (fun x -> e’)

evaluates to a threadt that first waits for threade to terminates.
The return value is then bound to parameterx in order to ex-
ecute threade’. The threadt behaves thereafter as threade’.
Intuitively, this can be read as alet expression for threads:
“let x = e in e’”. It is also a synchronization primitive: thread
e’ starts running only after threade is terminated. The thread
Lwt.return e is a thread that terminates immediately with value

1http://www.pps.jussieu.fr/~vouillon/lwt/



the value of expressione. The type of threads together with func-
tionsLwt.bind andLwt.return forms a monad.

In order to use the library, the functionforward is rewritten as
follows:

let forward in_fd out_fd buffer =
Lwt.bind

(Lwt_unix.read in_fd buffer 0 1024)
(fun n ->

Lwt.bind
(Lwt_unix.sleep 3)
(fun () ->

Lwt.bind
(Lwt_unix.write out_fd buffer 0 n)
(fun n’ ->

Lwt.return ())))

Thus, the function now returns immediately (that is, without block-
ing in a system call) a thread of typeunit Lwt.t. This thread first
reads up to one kilobyte. Whenn bytes have been read, the thread
sleeps for three seconds. Then, it writes then bytes. Finally, it re-
turns the unit value.

The syntax can be made more readable by using a binary oper-
ator forLwt.bind:

let (>>=) = Lwt.bind

Then, the function can be rewritten as follows.

let forward in_fd out_fd buffer =
Lwt_unix.read in_fd buffer 0 1024 >>= fun n ->
Lwt_unix.sleep 3 >>= fun () ->
Lwt_unix.write out_fd buffer 0 n >>= fun n’ ->
Lwt.return ()

The second example is self-contained. It presents the thread
scheduler. We first define a functionloop. This function returns
a thread printing a text each second.

let rec loop s =
Lwt_unix.sleep 1 >>= fun () ->
Format.printf "Hello %s@." s;
loop s

Next, two threads are created. This is done implicitly: threads starts
executing at once and there is no need to call a function such as
spawn to launch them. The threads are immediately suspended as
they callLwt_unix.sleep. Then, we enter the scheduler, which
takes care of resuming the appropriate threads when a systemcall
terminates: the expressionLwt_unix.run e executes the sched-
uler until threade terminates. Here, the scheduler runs for ever as
the threadLwt.wait () is a thread that never terminates.

let _ =
let thread1 = loop "A" in
let thread2 = loop "B" in
Lwt_unix.run (Lwt.wait ())

3. The library API
The library has two main components, which we present in turn.
The moduleLwt provides the core functionalities. The module
Lwt_unix provides a scheduler and gives access to the Unix I/O
system calls. Some additional modules implement functionalities
such as locks ant timeouts.

3.1 The core library

The interface of moduleLwt is given in figure 1. The type’a t
is the type of threads which, if they terminate successfully, yield a

type ’a t

val return : ’a -> ’a t
val bind : ’a t -> (’a -> ’b t) -> ’b t
val fail : exn -> ’a t
val catch :
(unit -> ’a t) -> (exn -> ’a t) -> ’a t

val try_bind :
(unit -> ’a t) ->
(’a -> ’b t) -> (exn -> ’b t) -> ’b t

val wait : unit -> ’a t
val wakeup : ’a t -> ’a -> unit
val wakeup_exn : ’a t -> exn -> unit
val poll : ’a t -> ’a option

Figure 1. The signature of moduleLwt

value of type’a. A thread can also run for ever, or fail with some
exception.

The threadreturn e is a thread that terminates immediately
with return value the value of expressione. The threadbind t f is
a thread which first waits for the completion of threadt and then, if
the thread succeeds, behaves as the application of functionf to the
return value of threadt. If the threadt fails, bind t f also fails,
with the same exception. As mentioned in section 2, it is convenient
to use the binary operator>>= as an alias for functionbind.

The typet together with functionbind and return forms a
monad. The following laws are satisfied:

return e >>= fun x -> t ≡ let x = e in t

t >>= fun x -> return x ≡ t

(t >>= fun x -> t’) >>= fun y -> t’’
≡

t >>= fun x -> (t’ >>= fun y -> t’’)

These are the monad laws adapted to the effectful semantics of ML.
The threadfail e is a thread that fails immediately with

exceptione. The threadcatch (fun () -> t) f behaves as
threadt as long as this thread does not fail. If the thread fails with
some exceptione, it behaves thereafter as the thread resulting from
applying functionf to exceptione. Functionsfail and catch
satisfy the following laws:

fail e >>= fun x -> t ≡ fail e

catch (fun () -> fail e) (fun x -> t)
≡

let x = e in t

catch (fun () -> return e) (fun x -> t)
≡

return e

For the sake of convenience, the functioncatch also captures ex-
ceptions raised by the usual ML exception mechanism. For in-
stance, in the following expression, the exceptionExit is caught
and applied to functionh.

catch (fun () -> raise Exit) h

The functiontry_bind is a generalization of functionsbind
andcatch, as proposed in (Benton and Kennedy 2001). The thread
try_bind (fun () -> t) f g behaves asbind t f if threadt
does not fail (and no exception is raised during its evaluation). It
behaves ascatch t g otherwise.

The last four functions are less useful for the casual user. They
provide the basic primitives on which can be built multi-threaded



libraries such asLwt_unix. The threadwait () is a thread which
is suspended, possibly forever. The functionwakeup terminates
such a thread with a given return value. The functionwakeup_exn
makes it fail with a given exception. The expressionpoll t tests
the state of threadt. It returnsSome v if threadt is terminated and
returned the valuev. If the thread failed with some exception, this
exception is raised. If the thread is still running, the value None is
returned without blocking.

3.2 The Lwt Unix module

This module provides all functionalities required to perform Unix
I/O system calls asynchronously. The module implements itsown
type of file descriptors. This ensures that all file descriptors used
in the library are in non-blocking mode, which is mandatory for
the library to work properly. The library also ensures that,once a
descriptor is closed, any attempt to access it fails. This isnot the
case when using a standard Unix library, where a same descriptor
can be reused for another file. This provides a protection against
programming errors with thread continuing to write to a closed
descriptor and ending up writing to the wrong file. Conversion
functions from and to the file descriptors of the standardUnix
library are provided.

The module exports the usual I/O functions such asread,
write or connect corresponding to the ones in theUnix library.
The types of functions that do not block such asclose or listen
are unchanged:

val close : file_descr -> unit

Possibly blocking functions have their types changed to return a
thread:

val read : file_descr ->
string -> int -> int -> int Lwt.t

The module also implements a scheduler. The role of the sched-
uler is to repeatedly wait for events (timeouts, file descriptors ready
for read or write, ...) and to resume the appropriate threads. The
scheduler is entered by calling functionrun of type

’a Lwt.t -> ’a.

The expressionrun t executes threadt (and other concurrent
threads) until it terminates. The expression evaluates to the return
value of the thread. If the thread fails with some exception,this
exception is raised. A thread can suspend itself temporarily to let
other threads run by calling functionyield.

Finally, the functionabort of type

file_descr -> exn -> unit

aborts all current and future operations on the given file descriptor
with the given exception (except for functionclose so that the file
descriptor can be properly closed). This provides a convenient way
to interrupt a thread that loops reading on or writing to a given file
descriptor, for instance after some timeout.

3.3 Additional modules

The following modules are also available:

• the moduleLwt_chan provides buffered I/O functions, similar
to the ones in modulePervasive of the Objective Caml stan-
dard library;

• the moduleLwt_mutex implements mutual exclusion locks
(see section 4.3 for details);

• the moduleLwt_timeout provides a timeout mechanism; it
can be used for instance withLwt_unix.abort to close a
network connection after some idle time.

• the moduleLwt_preemptive allows to mix preemptive threads
with Lwt cooperative threads. It maintains an extensible pool of
preemptive threads on with computations can be performed.

3.4 Pitfalls

While using the library, we have noticed a number of pitfallsthat
we document here.

First, it is mandatory to use the functionscatch or try_bind
for handling thread exceptions rather than the usual exception han-
dling mechanism: an expression

try t with e -> raise Some_Exception

will not catch any exception embedded in threadt. This is a point
that one must keep in mind when converting existing code to use the
library. Indeed, the type checker properly force us to usebind and
return wherever needed but the expression above remains well-
typed whatever the type of expressiont.

Second, the programmer expects uncaught exceptions to termi-
nate the program. This is not the case with an exception embed-
ded in a thread. For instance, the expressionfail e has no di-
rect effect. Thus, an exception handler should always be explicitly
wrapped around any thread that may fail with an exception, sothat
the exception is not silently ignored.

Last, it is very tempting to call functionLwt_unix.run to get
the result of a thread without using functionbind when not in a
thread context, given the type of this function:

’a Lwt.t -> ’a.

This works properly only as long as the call is not made from a
thread. Calling this function from a thread may result in a deadlock,
as it introduces spurious synchronizations between threads. Indeed,
call to functionrun are regular function calls and thus must be
properly nested. Thus, if two threads call functionrun, the first
thread cannot exit from the function before the second one.

4. Using the library
The following three larger examples illustrate the use of the library.
The first one, a port forwarder (section 4.1), is a typical example
of network programming. The two other ones, a simple scheduler
(section 4.2) and an implementation of locks (section 4.3),demon-
strate the expressivity of the short API of moduleLwt.

4.1 Network programming

We present the implementation of a port forwarder. The program
waits for connections on port 8080. When a client connects tothis
port, a remote connection to port 80 ongoogle.com is established.
Any data received on either side is then forwarded to the other side.

The auxiliary functionreally_write write l bytes of strings
starting at positionp to file descriptoro. Several calls to function
Lwt_unix.write may be needed as the system call may write less
thanl bytes.

let rec really_write o s p l =
Lwt_unix.write o s p l >>= fun n ->
if l = n then

Lwt.return ()
else

really_write o s (p + n) (l - n)

The functionforward writes everything it reads on file descriptori
to file descriptoro. When theLwt_unix.read system call returns
zero, which indicates end of file, we shut down the sending part of
the other connection and stop transmitting.

let rec forward i o =
let s = String.create 1024 in



Lwt_unix.read i s 0 1024 >>= fun l ->
if l > 0 then begin

really_write o s 0 l >>= fun () ->
forward i o

end else begin
Lwt_unix.shutdown o Unix.SHUTDOWN_SEND;
Lwt.return ()

end

The functionnew_socket creates a TCP socket. The program ex-
pects connections on addresslocal_addr and establish connec-
tions to addressremote_addr.

let new_socket () =
Lwt_unix.socket

Unix.PF_INET Unix.SOCK_STREAM 0
let local_addr =
Unix.ADDR_INET (Unix.inet_addr_any, 8080)

let remote_addr =
let host_entry =

Unix.gethostbyname "google.com" in
let inet_addr =

host_entry.Unix.h_addr_list.(0) in
Unix.ADDR_INET (inet_addr, 80)

The functionaccept returns a thread that repeatedly accept con-
nections on the socketsock. The thread blocks until a connec-
tion is established. Then, a thread dealing with the connection is
started asynchronously (parenthesized expression) and the function
is called recursively to wait for another connection. The thread con-
nects to the remote address. It starts two threads for forwarding
data in both directions. It then waits for the two threads to termi-
nate and finally closes the file descriptors corresponding toboth
connections.

let rec accept sock =
Lwt_unix.accept sock >>= fun (inp, _) ->
ignore

(let out = new_socket () in
Lwt_unix.connect

out remote_addr >>= fun () ->
let io = forward inp out in
let oi = forward out inp in
io >>= fun () -> oi >>= fun () ->
Lwt_unix.close out;
Lwt_unix.close inp;
Lwt.return ());

accept sock

We can now write the body of the program: it creates a socket that
listens on the local address and starts accepting connections.

let _ =
let socket = new_socket () in
Lwt_unix.setsockopt

socket Unix.SO_REUSEADDR true;
Lwt_unix.bind socket local_addr;
Lwt_unix.listen socket 1024;
Lwt_unix.run (accept socket)

Clearly, a robust version of this program should catch Unix er-
rors so as to properly shutdown the connections. Besides, writ-
ing to a socket whose other end is closed results by default ina
SIGPIPE signal. This signal should be ignored by the program so
thatLwt_unix.write fails with anEPIPE error instead.

4.2 A simple scheduler

We present a simple scheduler. The implementation of the more
complete scheduler provided by moduleLwt_unix is sketched

in section 5.6. Here, a thread can temporarily pause by calling a
functionyield, in order to allow other threads to execute. The task
of this scheduler is to then restart another thread. The scheduler is
started by calling a functionrun.

We first define the FIFO queue of suspended thread, using the
Queue module of Objective Caml standard library.

let queue = Queue.create ()

The scheduler repeatedly takes a thread from the queue and re-
sumes it. The thread runs until a call toyield, which gives back
the control to the scheduler which can then restart another thread.
The scheduler stops when the queue becomes empty.

let rec run () =
match

try
Some (Queue.take queue)

with Queue.Empty ->
None

with
None -> ()

| Some t -> Lwt.wakeup t (); run ()

The functionyield creates a suspended thread, adds the thread to
the end of the queue and returns it. Thus, a thread waiting foran
expressionyield () to terminate is stopped until resumed by the
scheduler.

let yield () =
let res = Lwt.wait () in
Queue.push res queue;
res

Here is an example of use of the scheduler. Theloop function
printsn times a strings, letting other threads run at each iteration.

let rec loop s n =
if n > 0 then begin

Format.printf "%s@." s;
yield () >>= fun () ->
loop s (n - 1)

end else
Lwt.return ()

Two threads are started, that output alternatively"a" and"b".

let _ =
let ta = loop "a" 6 in
let tb = loop "b" 5 in
run ()

4.3 Mutexes

We present an implementation of mutual exclusion locks. It follows
the corresponding implementation in the Objective Caml standard
thread library. Interestingly, this implementation only depends on
moduleLwt. It can thus be used with any scheduler. A mutex is a
pair of a boolean which indicates whether the mutex is lockedand
a list of threads waiting for the mutex to become unlocked.

type t =
{ mutable locked : bool;

mutable waiting : unit Lwt.t list }

A mutex is initially created unlocked.

let create () =
{ locked = false; waiting = [] }

The functionlock attempts to lock a mutexm. If the mutex is not
locked, it is locked and the function returns immediately. Other-
wise, the thread is suspended: a suspended thread is createdand



added to the list of waiting threads. The function waits for this
thread to be resumed and then calls itself recursively.

let rec lock m =
if not m.locked then begin

m.locked <- true;
Lwt.return ()

end else begin
let res = Lwt.wait () in
m.waiting <- res :: m.waiting;
res >>= fun () ->
lock m

end

In order to release a mutex, a copy of the set of waiting threads w
is extracted and the set is cleared. The mutex is then unlocked.
Finally, the waiting threads are restarted. It is importantto perform
this step last in order to avoid race conditions.

let unlock m =
let w = m.waiting in
m.waiting <- [];
m.locked <- false;
List.iter (fun t -> Lwt.wakeup t ()) w

5. Implementation
We first present in details the core moduleLwt. The implementation
of moduleLwt_unix is then sketched in section 5.6. The code
presented here deviates slightly from the actual implementation for
the sake of readability.

5.1 The type of threads

Threads are represented by a memory cell with a mutable state.

type ’a t =
{ mutable state : ’a state }

and ’a state =
Return of ’a

| Fail of exn
| Sleep of (’a t -> unit) list ref
| Link of ’a t

The three main states are:

• Return v: the thread has terminated successfully with the
valuev;

• Fail e: the thread has failed with exceptione;

• Sleep w: the thread is not finished yet; the thunk functions in
setw are called when the thread terminates.

The last stateLink t is used to implement a union-find datastruc-
ture2 over threads in order to coalesce threads with an identical
behavior. This turns out to be crucial to avoid some memory leaks
(see section 5.5.2 for details). Thefind function returns the rep-
resentative of a thread.

let rec find t =
match t.state with

Link t’ -> find t’
| _ -> t

The actual implementation of functionfind uses path compres-
sion: each visited thread gets directly linked to its representative.

5.2 Creating a thread

There is a function for creating a thread in each state.

2http://en.wikipedia.org/wiki/Union-find

let return v = { state = Return v }
let fail e = { state = Fail e }
let wait () = { state = Sleep (ref []) }

5.3 Terminating a thread

We present the implementation of the two functionswakeup and
wakeup_exn. They both rely on a functionfinish that changes the
state of a thread from still running to terminated. Its arguments are
the threadt and the new statest (which should be eitherReturn v
or Fail e). First, the representative of the thread is found. Then,
the list of waiters is extracted and the state of the thread ischanged
to st. Finally, the waiters are awaken. It is crucial to perform this
step last in order to avoid a race condition where new waitersare
added while processing current waiters.

let finish t st =
let t = find t in
match t.state with

Sleep waiters ->
t.state <- st;
List.iter (fun f -> f t) !waiters

| _ ->
invalid_arg "finish"

The implementation of functionswakeup andwakeup_exn is now
straightforward.

let wakeup t v = finish t (Return v)
let wakeup_exn t e = finish t (Fail e)

5.4 Thread synchronization

We present the implementation of functionsbind, try_bind and
catch. These functions make use of a functiontrap for catching
ML exceptions and embed them into a failing thread.

let trap f x = try f x with e -> fail e

They also rely on a functionconnect of type:

’a t -> ’a t -> unit.

A call connect t t’, where threadt must not be terminated,
ensures that the behavior of threadt mimics thereafter the behavior
of threadt’: threadt will terminate when threadt’ terminates,
with the same result. The easy case is whent’ is already finished.
Then, threadt is terminated with the same state ast’ (call to
function finish). Otherwise, the threads are both still running.
Then, the representative oft’ is linked to the representative oft
and the waiter sets are merged. The actual implementation uses
lists with constant-time append in order to make the cost of this
last operation independent from the number of waiters.

let rec connect t t’ =
let t’ = find t’ in
match t’.state with

Sleep waiters’ ->
let t = find t in
begin match t.state with

Sleep waiters ->
waiters := !waiters’ @ !waiters;
t’.state <- Link t

| _ ->
invalid_arg "connect"

end
| _ ->

finish t t’.state

All three synchronization functions share a common core, function
try_bind_rec. This function takes as argument a threadt and
two functionsf and g. If t is terminated with valuev, then the



application of functionf to valuev is returned. Ift has failed with
exceptione, then the application of functiong to exceptione is
returned. Ift is not yet terminated, a fresh suspended threadres is
created. A thunk is added to the set of waiters of threadt so that,
when threadt terminates, the functiontry_bind_rec is called
again and the behavior ofres follows the behavior of the thread
returned by this function. The threadres is finally returned. The
functiontry_bind_rec is called recursively at most once, ast is
always terminated when the recursive call is performed.

let rec try_bind_rec t f g =
match (find t).state with

Return v ->
f v

| Fail e ->
g e

| Sleep waiters ->
let res = wait () in
waiters :=
(fun t ->

connect res
(try_bind_rec t (trap f) (trap g)))

::
!waiters;

res
| Link _ ->

assert false

From this function, all three synchronization operators can easily
be written:

let bind t f = try_bind_rec t f fail

let try_bind f g h =
try_bind_rec (trap f ()) g h

let catch f g = try_bind f return g

The use of functiontrap in the functionstry_bind_rec and
try_bind is clarified in section 5.5.1

5.4.1 Polling for a thread state

The implementation of the function testing the state of a thread is
straightforward.

let poll t =
match (repr t).state with

Fail e -> raise e
| Return v -> Some v
| Sleep -> None
| Repr _ -> assert false

5.5 Implementation difficulties

The implementation has now been presented in full. The following
explains in more details some subtle issues.

5.5.1 Dealing with exceptions

The functionLwt.catch as used in the expression below will catch
not just the exceptions embedded in threadt (for instance, using
the functionfail) but also the exceptions raised by the usual ML
mechanism (using the operatorraise).

Lwt.catch (fun () -> t) handler

This is implemented by calling functiontrap at suitable places.
When the exception is raised during the evaluation of the thunk,
it is caught by the occurrence of this function in the body of
functiontry_bind. This covers the following example:

Lwt.catch
(fun () ->

raise Not_found;
Lwt.return ())

handler

The following second case, where functionLwt.bind is applied to
a thread which is already terminated, is also covered by the same
call to functiontrap. Indeed, the exception is raised before the
whole thunk function exits.

Lwt.catch
(fun () ->

Lwt.return () >>= fun () ->
raise Not_found;
Lwt.return ())

handler

In the last case below, the evaluation of the thunk results ina
suspended thread, and the exception is raised only after thethread
resumes. In this case, the exception is caught by the calls tofunction
trap in functiontry_bind_rec.

Lwt.catch
(fun () ->

Lwt_unix.yield () >>= fun () ->
raise Not_found;
Lwt.return ())

handler

One has to be careful about where the calls to functiontrap are
performed. In particular, it would be incorrect to protect systemat-
ically the second argument of functionbind, as this would break
tail-recursion. However, it is not an issue to do it after a recursive
call to functiontry_bind_rec, that is, in response to a thread ter-
mination. Indeed, this termination is generally performedby a call
to functionwakeup or functionwakeup_exn from the scheduler,
and the scheduler body is a tail-recursive function and is thus at a
fixed stack depth.

5.5.2 Avoiding memory leaks

It is crucial that the memory behavior of threads conforms tothe
expectations of the programmer. In particular, a function which is
tail-recursive when written in a non-threaded way should betail-
recursive when usingLwt. This is not the case with a naive imple-
mentation that does not coalesce equivalent threads. Indeed, con-
sider the following piece of code, using the scheduler in section 4.2.

let rec loop n =
if n = 0 then Lwt.return () else
yield () >>= fun () -> loop (n - 1)

in
let l = loop 100000 in
run ()

The function yield returns a suspended threadt. The func-
tion loop thus also initially returns a suspended threadt1. Later,
the threadt is successfully terminated by the scheduler. This trig-
gers a recursive call to functionloop returning a new threadt2

which is connected to threadt1 by functionconnect. This is re-
peated again and again, thus we get a longer and longer chain:

A straightforward implementation of functionconnect, when ap-
plied to threadsti andti+1, would add a thunk to the list of waiters
of threadti+1. The goal of this thunk is to update threadti when
threadti+1 is terminated. But then, as long as the head of the chain
remains live (that is, as long as the loop has not ended), noneof



the chain can be garbage collected. Our implementation usesthe
fact that all these threads behave the same: they are suspended until
the loop ends, and then are all terminated with value unit. Ittakes
threadt1 as the unique representative for them all. Then, all inter-
mediate threads can be garbage collected:

We conjecture that, with this implementation, translatingan exist-
ing code to useLwt does not introduce any memory leak.

5.6 The Unix library

We sketch the implementation of the scheduler and present the
implementation of a possibly blocking system call. For the sake
of clarity, we only present the way the scheduler deals with op-
erations on file descriptors. The actual implementation also deals
with threads to be restarted after a given amount of time (func-
tions sleep and yield) and subprocesses termination (function
waitpid).

The threads waiting for I/Os are stored in two datastructures
inputs and outputs that associate to some file descriptors the
actions that should be performed when they become availablere-
spectively for reading and writing.

The scheduler (functionrun below) loops until the input thread
thr terminates. The status of the thread is checked at each it-
eration by calling functionLwt.poll. If it is not finished yet,
the scheduler proceeds to call the functionUnix.select to wait
for file descriptors to become available. The listsinfds and
outfds of file descriptors to watch are computed by function
active_descriptors. The third list is for waiting for so-called
socket exceptions (that is, out-of-band data), which is notcurrently
supported (this socket feature is hardly ever used). The float -1.0
indicates that the wait is unbounded (no timeout). The system call
is interrupted with errorEINTR when a signal occurs, resulting in
an ML exception. This exception can be ignored. When the system
call returns, the corresponding actions are performed on each avail-
able descriptor by calling functionperform_action. Finally, the
scheduler calls itself recursively.

let rec run thr =
match Lwt.poll thr with

Some v ->
v

| None ->
let infds = active_descriptors inputs in
let outfds = active_descriptors outputs in
let (readers, writers, _) =

try
Unix.select infds outfds [] (-1.0)

with
Unix_error (Unix.EINTR, _, _) ->

([], [], [])
in
List.iter

(fun fd -> perform_actions inputs fd)
readers;

List.iter
(fun fd -> perform_actions outputs fd)
writers;

run thr

In the actual implementation, timeouts (functionsleep) are man-
aged by using a priority queue. A timeout value is given to function
select instead of the float-1.0 in order to interrupt the system

call when a thread has to be resumed. Child termination (func-
tion waitpid) is detected by catching the signalSIGCHILD.

The functionperform_actions eventually invokes the func-
tion wrap_syscall shown below to perform an actionaction
on file descriptorch. In order to be able requeue the action in
case it fails to complete, the function takes as additional argu-
ments the datastructureset that held the action and the thread
cont to be resumed when the action is completed. The func-
tion check_descriptor is called to check whether the function
abort was previously called on the file descriptor and raises the
corresponding exception if this is the case (see section 3.2). It
also raises an exception when the file descriptor is marked as
closed. The action is then attempted. If it fails with a Unix error
EAGAIN, EWOULDBLOCK or EINTR, the action is requeued (function
add_action). Otherwise, the thread is resumed.

let rec wrap_syscall set ch cont action =
let res =

try
check_descriptor ch;
Success (action ())

with
Unix.Unix_error

((Unix.EAGAIN | Unix.EWOULDBLOCK |
Unix.EINTR),_,_) ->

add_action set ch cont action;
Requeued

| e ->
Exn e

in
match res with

Success v ->
Lwt.wakeup cont v

| Exn e ->
Lwt.wakeup_exn cont e

| Requeued ->
()

We now show how the functionwrite is implemented. It first
checks the file descriptor. If this does not result in an exception, the
system callwrite is performed. The Unix errors indicating that the
write would block are caught and result in scheduling the write to
be attempted again when the file descriptor becomes available. This
is performed by calling functionregister_action which returns
a suspended thread that is resumed when the action completes. If
another error occurs, a failing thread is returned.

let write ch buf pos len =
try

check_descriptor ch;
Lwt.return (Unix.write ch.fd buf pos len)

with
Unix.Unix_error

((Unix.EAGAIN | Unix.EWOULDBLOCK |
Unix.EINTR), _, _) ->

register_action outputs ch
(fun () -> Unix.write ch.fd buf pos len)

| e ->
Lwt.fail e

6. Related work
The idea to implement cooperative threads using a monad is due
to Claessen (Claessen 1999). Li and Zdancewic have written an
implementation in Haskell (Li and Zdancewic 2007) with perfor-
mance in mind. They use the efficientepoll Linux mechanism in-
stead of the more portable but less efficientselect system call to



implement their scheduler. The company Liveops has developed a
similar monad-based library (Waterson 2007) for ObjectiveCaml.
They report that their library does not deal well with exceptions.
The library has not been publicly released yet.

These works all use some variants of a continuation monad,
which makes the semantics of their threads slightly different from
ours. The type of threads is typically similar to the following one:

’a t = (’a -> unit) -> unit

This is a functional type: an expression of this type does noth-
ing before being given a continuation. For instance, an expression
read fd buffer 0 512 does not attempt at once to read on file
descriptorfd as is the case withLwt. The read is only attempted
once a continuation is provided. Thread creation is then explicit:
a function, usually calledfork or spawn, must be called to apply
a thread to its final continuation and thus start the thread execu-
tion. As an expression of type’a t is not a running thread, the
function bind is just a sequencing operator and does not provide
a communication mechanism between threads. A separate mecha-
nism has to be provided.

An alternative to threads for highly concurrent network appli-
cations is event-based programming. Theequeue library, part of
the Ocamlnet (Stolpmann) library, is an Objective Caml library for
event queues. It is used by a number of other libraries in Ocamlnet
to parallelize network code. Compared toLwt, the API is very low-
level. It should be possible to built aLwt scheduler on top of this
library, so that code written usingLwt can interact with the library.

We have not performed any benchmark, but we believe that
our implementation is competitive performance-wise with respect
to other Objective Caml thread implementations. Indeed, they all
share the same limitation that only one thread is active at a given
time. An advantage of our library is the low cost of thread creation.

